

# Hoare logic and Model checking

## Part II: Model checking

### Lecture 10: Implementing model checking

---

**Hiroiyuki Katsura** hk590

University of Cambridge

CST Part II – 2025/26

In the last two lectures we saw LTL and CTL as examples of temporal logics that can specify the behaviour of temporal models. For using temporal logics in the verification of artefacts, we need model checkers to check a temporal model against a temporal logic specification.

In this lecture we will implement a naïve model checker for CTL: the world's worst model checker for CTL.

# Model checking

What is model checking?

The model checking problem for CTL is to determine, for a given a temporal model  $M$  over some set of atomic propositions  $AP$  and CTL formula  $\psi$  over  $AP$ , whether  $M$  satisfies  $\psi$ :

$$M \models \psi$$

We need a function that computes this.

## Definite temporal models

The temporal models we have defined in lecture 7 were not restricted in any way that guarantees computability. Here we will assume a **definite temporal model**, using a finite set of states and computable functions for the initial state predicate, the transition relation and the labelling of states.

## Type of temporal models

```
type state = int
```

```
module States = Set.Make(Int)
```

```
type 'ap tmodel = {  
  s : States.t; (* finite set *)  
  s0 : state -> bool; (* computable *)  
  t : state -> state -> bool; (* computable *)  
  l : state -> 'ap -> bool; (* computable *)  
}
```

## Specifying a CTL model checker

We will implement a naïve CTL model checker:

```
val mc : 'ap tmodel -> 'ap state_prop -> bool
```

which has the following specification:

$$\forall M, \psi. (\text{mc } M \psi \Leftrightarrow M \models \psi)$$

## Defining a CTL model checker

To check whether the model satisfies a property  $\psi$ , we have to check whether the initial states satisfy  $\psi$ . We check this using an auxiliary function *mca* that returns the states satisfying a given state property.

```
let mc (m : 'ap tmodel) (psi : 'ap state_prop) : bool =
  assert (left_total m);
  let v = mca m psi in
  States.for_all (fun s ->
    not (m.s0 s) || States.mem s v
  ) m.s
```

This *mca* function works by recursion on the formula, calling itself on the sub-formulas.

## CTL model checker

(This is often phrased in terms of “labelling” of states.)

**Strategy:** For a given CTL state-property  $\psi$ : compute the states of the temporal model that satisfies  $\psi$ , by

- exploiting CTL formula equivalences to encode  $\psi$  as a formula  $\hat{\psi}$  that uses only existential path quantification (using negation in the right places)
- (recursively) computing the states satisfying the **sub-formulas** of  $\hat{\psi}$ , and
- using this information to determine which states should be returned for  $\hat{\psi}$ .

## CTL model checker: propositional fragment

*mca*, for a given temporal model and state property returns the set of states satisfying the state property.

```
let rec mca (m : 'ap tmodel) (psi : 'ap state_prop)
  : States.t =
  match psi with
  | True ->
    m.s
  | False ->
    States.empty
  | AP p ->
    States.filter (fun s -> m.l s p) m.s
  | Not psi' ->
    let v = mca m psi' in
    States.diff m.s v
  ...
```

## CTL model checker: propositional fragment (continued)

```
let rec mca (m : 'ap tmodel) (psi : 'ap state_prop)
  : States.t =
  ...
  | And (psi1, psi2) ->
    let v1 = mca m psi1 in
    let v2 = mca m psi2 in
    States.inter v1 v2
  | Or (psi1, psi2) ->
    let v1 = mca m psi1 in
    let v2 = mca m psi2 in
    States.union v1 v2
  | Impl (psi1, psi2) ->
    mca m (Or (Not psi1, psi2))
  ...
```

## CTL model checker: A

We use

- $A X \psi' = \neg E X (\neg \psi')$
- $A G \psi' = \neg E F (\neg \psi')$

```
let rec mca (m : 'ap tmodel) (psi : 'ap state_prop)
  : States.t =
  ...
  | A (X psi') ->
    mca m (Not (E (X (Not psi'))))
  | A (G psi') ->
    mca m (Not (E (F (Not psi'))))
  | A (F _) ->
    failwith "TODO:␣exercise"
  | A (U (psi1, psi2)) ->
    failwith "TODO:␣tricky␣exercise"
  ...
```

## CTL model checker: EX

If we know in which states  $\psi'$  holds, then we know in which states  $X \psi'$  holds: their predecessors:

```
let rec mca (m : 'ap tmodel) (psi : 'ap state_prop)
  : States.t =
  ...
  | E (X psi') ->
    let v = mca m psi' in
    States.filter (fun s ->
      States.exists (fun s' ->
        m.t s s'
      ) v
    ) m.s
  ...
```

## CTL model checker: EF

We use  $E F \psi' = E (T U \psi')$

```
let rec mca (m : 'ap tmodel) (psi : 'ap state_prop)
  : States.t =
  ...
  | E (F psi') ->
    mca m (E (U (True, psi')))
  ...
```

## CTL model checker: EG and EU

Left to do are  $E G \psi'$  and  $E (\psi_1 U \psi_2)$ , which talk about infinite paths. We will implement those using fixpoint operations on sets, where the finite size of the set of states guarantees termination.

## Fixpoint.

```
let rec fixpoint (f : States.t -> States.t)
                (s : States.t) : States.t =
  let s' = f s in
  if States.equal s s' then s else fixpoint f s'
```

## CTL model checker: EG

For  $E G \psi'$ :

1. compute the set  $v$  of states satisfying  $\psi'$
2. define the output set to be  $v' := v$
3. until there are no more changes: remove from  $v'$  elements that cannot transition into  $v'$

```
| E (G psi') ->
```

```
  let v = mca m psi' in
  fixpoint (fun v' ->
    States.filter (fun s ->
      States.exists (fun s' ->
        m.t s s'
      ) v'
    ) v'
  ) v
```

...

## CTL model checker: EU

For E ( $\psi_1$  U  $\psi_2$ ):

1. compute the sets  $v_1$  and  $v_2$  of states satisfying  $\psi_1$  and  $\psi_2$
2. define the output set to be  $v' := v_2$
3. until there are no more changes: add states from  $v_1$  that can transition into  $v'$

```
| E (U (psi1, psi2)) ->  
  let v1 = mca m psi1 in  
  let v2 = mca m psi2 in  
  fixpoint (fun v' ->  
    States.union v'  
    (States.filter (fun s ->  
      States.exists (fun s' ->  
        m.t s s'  
      ) v'  
    ) v1)  
  ) v2
```

## Actually implementing model checking

This is not very efficient!

In practice,

- the labelling (the  $vs$ ) are memorised: in our code the  $vs$  are re-computed each time, in the case of nested CTL formulas
- “symbolic model checking” uses binary decision diagrams (IB Logic and proof) to represent sets of states, and performs operations on sets-as-BDDs, instead of explicitly manipulating the sets;
- the states can be computed lazily;
- “partial order reduction” tries to not enumerate redundant interleavings;
- ...
- 40+ years of tricks!

## Counterexamples

---

## Generating counterexamples

Adapted from “Tree-Like Counterexamples in Model Checking”.

If the specification is not satisfied, and is in ACTL, then we can do better than just say “no”: we can produce a counterexample.

The idea is that  $M \not\models \psi^{\text{ACTL}}$  is equivalent to  $M \models \neg\psi^{\text{ACTL}}$ , where  $\neg\psi^{\text{ACTL}}$  can be expressed in ECTL.

So  $M \not\models \psi^{\text{ACTL}}$  implies the existence of a witness for the corresponding ECTL property.

We will now assume formulas in negation normal form: formulas without implication, and where the only use of negation is immediately preceding an atomic proposition.

## Shape of ECTL witnesses

The shape of an ECTL witness for a set of atomic propositions  $AP$  and temporal model  $M$ :

Witness $_M :=$

- | WAP  $\in M.S \rightarrow$  Witness $_M$
- | WNAP  $\in M.S \rightarrow$  Witness $_M$
- | WAnd  $\in$  Witness $_M \rightarrow$  Witness $_M \rightarrow$  Witness $_M$
- | WOrL  $\in$  Witness $_M \rightarrow$  Witness $_M$
- | WOrR  $\in$  Witness $_M \rightarrow$  Witness $_M$
- | WX  $\in M.S \rightarrow M.S \rightarrow$  Witness $_M \rightarrow$  Witness $_M$
- | WF  $\in$  list  $M.S \rightarrow$  Witness $_M \rightarrow$  Witness $_M$
- | WG  $\in$  list  $(M.S \times$  Witness $_M) \rightarrow$  Witness $_M$
- | WU  $\in$  list  $(M.S \times$  Witness $_M) \rightarrow$  Witness $_M$

There are (on purpose) no cases for A ...

## Being an ECTL witness

We will define when a witness is a “valid witness” for an ECTL property:

$(s \models_M \psi)$  wit-by  $W$

should hold whenever  $W$  is a valid witness for the fact that  $\psi$  holds in state  $s$  of temporal model  $M$ .

## Being an ECTL witness: atomic propositions

A witness for an atomic proposition is just the fact that the atomic proposition holds according to  $M \cdot \ell$ :

$$(s \models_M p) \text{ wit-by } W \stackrel{\text{def}}{=} \\ W = \text{WAP } s \wedge M \cdot \ell \ s \ p$$

Similarly for negation of atomic propositions.

## Being an ECTL witness: next

A witness for 'next' is a transition from the current state to a next state, and a witness that the sub-property holds in the next state:

$$(s \models_M (E X \psi)) \text{ wit-by } W \stackrel{\text{def}}{=} \\ \exists s' \in M.S, W' \in \text{Witness}_M. \\ \left( \begin{array}{l} W = WX \ s \ s' \ W' \wedge \\ s \ M.T \ s' \wedge \\ (s' \models_M \psi) \text{ wit-by } W' \end{array} \right)$$

## Being an ECTL witness: future

A witness for the 'future' temporal operator is a finite path that leads to a state for which we have a witness that it satisfies the sub-property:

$$(s \models_M E F \psi) \text{ wit-by } W \stackrel{\text{def}}{=} \\ \exists s' \in M.S, \pi \in \text{list } M.S, W' \in \text{Witness}_M. \\ \left( \begin{array}{l} W = WF \ \pi \ W' \wedge \\ \text{IsFinitePath } M \ \pi \wedge \\ \text{nth } \pi \ 0 = s \wedge \\ \text{last } \pi = s' \wedge \\ (s' \models_M \psi) \text{ wit-by } W' \end{array} \right)$$

## Being an ECTL witness: generally

A witness for the ‘generally’ temporal operator is a lasso-shaped path, together with witnesses that each state along the path satisfies the sub-property:

$$(s \models_M E G \psi) \text{ wit-by } W \stackrel{\text{def}}{=} \exists SWs \in \text{list } (M.S \times \text{Witness}_M). \left( \begin{array}{l} W = WG \ SWs \wedge \\ \text{let } \pi = \text{firsts } SWs \text{ in} \\ \text{IsFinitePath } M \ \pi \wedge \\ \text{nth } \pi \ 0 = s \\ (\exists i \in \mathbb{N}. (\text{last } \pi) \ M.T \ (\text{nth } \pi \ i)) \wedge \\ \left( \forall j \in \mathbb{N}, s' \in M.S, W' \in \text{Witness}_M. \right. \\ \left. \left( \begin{array}{l} \text{nth } SWs \ j = \langle s', W' \rangle \Rightarrow \\ (s' \models_M \psi) \text{ wit-by } W' \end{array} \right) \right) \end{array} \right)$$

## Being an ECTL witness: until

$$(s \models_M E (\psi_1 U \psi_2)) \text{ wit-by } W \stackrel{\text{def}}{=} \\ \exists SWs \in \text{list } (M.S \times \text{Witness}_M), s' \in M.S, W' \in \text{Witness}_M. \\ \left( \begin{array}{l} W = WU (SWs \text{ ++ } [\langle s', W' \rangle]) \wedge \\ \text{let } \pi = \text{firsts } SWs \text{ ++ } [s'] \text{ in} \\ \text{IsFinitePath } M \pi \wedge \\ \text{nth } \pi \ 0 = s \wedge \\ \left( \forall i \in \mathbb{N}, s'' \in M.S, W'' \in \text{Witness}_M. \right. \\ \left. \left( \begin{array}{l} \text{nth } SWs \ i = \langle s'', W'' \rangle \Rightarrow \\ (s'' \models_M \psi_1) \text{ wit-by } W'' \end{array} \right) \right) \wedge \\ \left( (s' \models_M \psi_2) \text{ wit-by } W' \right) \end{array} \right)$$

## Being an ECTL witness: conjunction

$$\begin{aligned} (s \models_M \psi_1 \wedge \psi_2) \text{ wit-by } W &\stackrel{\text{def}}{=} \\ \exists W_1 \in \text{Witness}_M, W_2 \in \text{Witness}_M. & \\ \left( \begin{array}{l} W = \text{WAnd } W_1 \ W_2 \wedge \\ (s \models_M \psi_1) \text{ wit-by } W_1 \wedge (s \models_M \psi_2) \text{ wit-by } W_2 \end{array} \right) & \end{aligned}$$

## Being an ECTL witness: disjunction

$(s \models_M \psi_1 \vee \psi_2) \text{ wit-by } W \stackrel{\text{def}}{=}$

$\exists W' \in \text{Witness}_M.$

$$\left( \begin{array}{l} \left( W = \text{WOrL } W' \wedge (s \models_M \psi_1) \text{ wit-by } W' \right) \vee \\ \left( W = \text{WOrR } W' \wedge (s \models_M \psi_2) \text{ wit-by } W' \right) \end{array} \right)$$

## Satisfiability and existence of witnesses

Here we have required finite temporal models, and so witnesses are finite. (Otherwise, we would need to deal with infinite witnesses.)

Now, if we have  $M \not\models \psi$  for some ACTL formula  $\psi$ , there exists a witness  $W$  for the fact that the ECTL formula corresponding to  $\neg\psi$  holds — and we could effectively find it by tweaking our model checking algorithm (details elided).

## Witnesses beyond ECTL

Can we have witnesses for more than just ECTL?

Yes. For example, one of the nice things about LTL is that counterexamples are just paths.

## Summary

We saw a model checking algorithm for CTL, and sketched how it could be modified to generate counterexamples for ACTL formulas.